
ABSTRACT: 
 
This article introduces a simple and useful method to directly apply the laboratory normal ranges data into 
the SAS program of determining the normality of the results of laboratory examinations. The procedure 
reads the normal ranges data from a separate raw data file and converts them into a SAS output text, 
which is combined with the conditional code to create an output program. The SAS output program is 
saved as a source code file and can be included in the major program of handling the laboratory data with 
normality determination. 
 
INTRODUCTION: 
 
It is a common practice in clinical trials to determine the normality of the laboratory examinations according 
to the predefined laboratory ranges. Usually, determining the lab-abnormality is a considerable part of 
laboratory data manipulation. The importance of finding the abnormal outcomes of laboratory 
examinations include: helping to clean the data for mistakenly entered unreasonable values; helping to 
check the qualification of the patients; and providing information for monitoring toxicity. It may also be used 
for concluding the efficacy of treatment, if laboratory results are the interested research outcomes.   
 
Most of the phase III clinical trials are multiple center trials. The trials are usually conducted in many 
clinical sites. The laboratory examinations can consequently be performed in different laboratories, using 
different methods and applying different concepts for normality. Some normal range criteria also vary with 
age and gender.  Therefore, determining normality of the results of lab examinations is time and resource 
consuming and could be an error-generating process. A robust, simple and valid approach, which 
implements laboratory normal ranges from different labs and is valid for different methods, age groups and 
gender, is of great importance in the lab data management. 
 
NORMAL RANGES DATA AND DIRECTLY USE THE DATA   
 
The general way of managing the laboratory normal ranges information involves collecting the data from 
different bio-analytical centers or laboratories (the data include lab names, the names of the examinations, 
and the normal criteria of the examination by age group and gender), and pooling all the data into a normal 
ranges data. The data contain the variables of lab-number, examination code (name), gender, the lower 
and upper age group boundaries, and the lower and upper value-limits of the examination results. In our 
practice, the normal ranges data are saved as spreadsheet data. It is convenient to manage and validate 
the data. An example of such data is shown below (see next page): 
 
It is a tiring and boring job to apply all of the normal criteria by typing them into the program of laboratory 
data manipulation. Given the normal ranges data exist, we can use the data directly into the lab data 
managing program. We can dump the values from the normal range data directly as the criteria in the 
laboratory normality determination program. 
 
THE SAS PROGRAM TO CREATE THE NORMALITY CHECK PROGRAM 
 
The simplified example program labnorm.sas, on the next page, uses DATA _null_ and conditional PUT 
statements to create the output program. It reads in the laboratory normal ranges data and combines the 
codes and the variable names (implementing their values) in the output file. The output file is a transitional 
SAS program. It is saved under the designated location. This file is actually the section of conditionally 
determining the normality for all the lab examinations in the lab data management program.   

 The example program 
 

 
The output program is named as nmrange.sas in the example (see next page). It consists of two sections: 
the high/low criteria determining (embedding criteria values in the program code) and the abnormality 
defining (the last section, which defines 1 as normal and 2 as abnormal).  
 

In the laboratory data management program, include the generated program in the data process and the 
values that are out of the predefined ranges for all the examinations will be flagged out. For a small testing 
data the patients with abnormal laboratory findings are listed in a report for clinical reviewing.    

An example lab ranges data  
      Age groups Normal Range
Lab no  test gender agel ageu Lower Upper

1 1 1   62 120
1 1 2   55 115
1 2 1   35 50
1 2 2   35 50
1 3 1   2.5 7
1 3 2   2.5 7
1 4 1   3.4 11
1 4 2   3.4 11
1 5 1   18 113
1 5 2   18 113

… … … … … … …
2 1 1   70 130
2 1 2   50 110
2 2 1   35 50
2 2 2   35 50
2 3 1   3 7
2 3 2   3 7
2 4 1   3.6 6
2 4 2   3.6 6
2 5 1   0 115
2 5 2   0 115

… … … … … … …
3 1 1   40 130
3 1 2   40 130
3 2 1   33 53
3 2 2   33 53
3 3 1   3 7
3 3 2   3 7
3 4 1   3.5 6
3 4 2   3.5 6
3 5 1 20 25 36 92
3 5 1 26 35 36 98
3 5 1 36 45 36 110
3 5 1 46 55 36 125
3 5 1 56 65 36 143
3 5 1 66 75 36 145
3 5 1 76  36 204
3 5 2 20 25 36 88
3 5 2 26 35 36 90
3 5 2 36 45 36 101
3 5 2 46 55 36 109
3 5 2 56 65 36 119
3 5 2 66 75 36 155
3 5 2 76  36 179

… … … … … … …

       

/***************************************
| labnorm.sas: a sas program to create  
|    code from the existing data for lab  
|    normality determination 
| DLI modified for SUGI 29. Dec.03 
****************************************/ 
 
options nodate nonumber nocenter  
        ps=999 ls=120; 
 
filename labn 'c:\dli\labnorm\labnorm.txt'; 
 
data labnorm; 
  infile labn DSD missover dlm='09'x irstobs=4;
  input Labno test gender agel ageu Lower    
        Upper; 
run; 
 
 
*** create code from data ***; 
 
data _null_; 
  set labnorm; 
  file "c:\dli\labnorm\nmrange.sas"; 
  if agel^=. and ageu^=. then do; 
    put "if labno=" labno  
        "and test=" test 
        "and gender=" gender 
        "and ("agel "<=age<=" ageu  
        ") then do; low=" lower  
        "; high=" upper"; end;"; 
  end; 
  if agel^=. and ageu=. then do; 
    put "if labno=" labno 
        "and test=" test 
        "and gender=" gender  
  "and age>=" agel 
        "then do; low=" lower  
        "; high=" upper"; end;"; 
  end; 
  if agel=. and ageu=. then do; 
    put "if labno=" labno 
        "and test=" test 
        "and gender=" gender  
        "then do; low=" lower  
        "; high=" upper"; end;"; 
  end; 
run; 
 
data _null_; 
  file "c:\dli\labnorm\nmrange.sas" MOD; 
put /; 
put "if value^=. and (value<low or value >high) 
    then normal=2; else normal=1;" ; 
run;

Selected code of output program  

  
IMPLEMENTING THE PROGRAM 
To include the generated program:   The abnormal lab-report looks like: 
 

data labnm; 
  set labdata; 
 
  %inc "c:\dli\labnorm\nmrange.sas"; 
run; 
 
proc print data=labnm; 
  where normal=2; 
title ‘Lab Abnormality Findings’; 
run; 

 
 
DISCUSSION AND CONCLUSION: 
 
If anyone thinks that this approach is so simple, think about how simple it was for Columbus to find the 
North American continent! The question here is: Does it work and can it simplify our job? It does and it 
can. Please note that as examples, the normal ranges file and lab data file are simple in this article. The 
real situation may be much more complicated. The process of lab data management can be remarkably 
simplified if the laboratory normal ranges data are directly used to make a transitional output program to 
be executed later in the data step. If any changes happen to the criteria, only the normal ranges data need 
to be modified. The program works in the same way. 
 
Furthermore, we can analogize the strategy to other similar situations, for example, automatically creating 
programming shells from a logic checking plan data. There are a lot of possibilities for directly using data 
to output program, when it applies.   

Other brand and product names are trademarks of their respective companies.  
 

if labno=1 and test=1 and gender=1 then do; low=62 ; high=120 ; end; 
if labno=1 and test=1 and gender=2 then do; low=55 ; high=115 ; end; 
if labno=1 and test=2 and gender=1 then do; low=35 ; high=50 ; end; 
if labno=1 and test=2 and gender=2 then do; low=35 ; high=50 ; end; 
 
 (more program lines …) 
 
if labno=3 and test=5 and gender=2 and (46 <=age<=55 ) then do;  
  low=36 ; high=109 ; end; 
if labno=3 and test=5 and gender=2 and (56 <=age<=65 ) then do;  
  low=36 ; high=119 ; end; 
if labno=3 and test=5 and gender=2 and (66 <=age<=75 ) then do;  
  low=36 ; high=155 ; end; 
if labno=3 and test=5 and gender=2 and age>=76 then do; low=36 ;  
  high=179 ; end; 
 
if value^=. and (value<low or value >high) then normal=2; else normal=1; 

Laboratory Abnormality Findings 
ptno  Labno  test  gender  age  value  low   high  normal 
 
 X01     1        2        1       30     55     35.0      50       2 
 X01     1        5        1       30    150    18.0    113      2 
 Y01     2        2        1       46     25     35.0      50      2 
 Y01     2        4        1       46       9       3.6       6       2 
 Z01     3        2        2       55      60     33.0     53      2 
 Z01     3        5        2       55    110     36.0   109      2 
 Z01     3        4        1       80       3       3.5       6       2 

